SOURCE CODE FOR THE COMPUTER PROGRAM AND SAMPLE DATA SET FOR THE SIMULA TION OF CYLINDRICAL FLOW T O A WELL USING THE U.S. GEOLOGICAL SUR VEY MODULAR FINITE-DIFFERENCE GROUND-W ATER FLOW MODEL

______________________________________________________________________________

U.S. GEOLOGICAL SURVEY

Open-File Report 92-659

2

SOURCE CODE FOR THE COMPUTER PROGRAM AND SAMPLE DATA SET FOR THE SIMULA TION OF CYLINDRICAL FLOW T O A WELL USING THE U.S. GEOLOGICAL SUR VEY MODULAR FINITE-DIFFERENCE GROUND-W ATER FLOW MODEL

by Thomas E. Reilly and Arlen W. Harbaugh

______________________________________________________________________________

U.S. GEOLOGICAL SURVEY

Open-File Report 92-659

Reston, Virginia 1993

ii 3

U.S. DEPARTMENT OF THE INTERIOR

BRUCE BABBITT, Secretary

U.S. GEOLOGICAL SURVEY

Dallas L. Peck, Director

______________________________________________________________________________

For additional information Copies of this report may be write to: purchased from:

U.S. Geological Survey Books and Open-File Reports Section National Center, MS 431 U.S. Geological Survey 12201 Sunrise Valley Drive Federal Center, Building 810 Reston, Virginia 22092 Box 25425 Denver, Colorado 80225

iii 4

CONTENTS

Abstract........................................................ 1

Introduction......................................................1

References...................................................... 2

Appendixes...................................................... 2

Appendix 1 - Program listing..................................... 3

Appendix 2 - Preprocessor input instructions........................... 6

Appendix 3 - Preprocessor input data for sample problem.................. 7

Diskette....................................................in pocket

iv 1

SOURCE CODE FOR THE COMPUTER PROGRAM AND SAMPLE DATA SET

FOR THE SIMULATION OF CYLINDRICAL FLOW TO A WELL USING THE

U.S. GEOLOGICAL SURVEY MODULAR FINITE-DIFFERENCE

GROUND-WATER FLOW MODEL by Thomas E. Reilly and Arlen W. Harbaugh

ABSTRACT

Cylindrical (axisymmetric) flow to a well is an important specialized topic of ground-water hydraulics and has been applied by many investigators to determine aquifer properties and determine heads and flows in the vicinity of the well. A recent modification to the U.S. Geological Survey Modular Three-Dimensional Finite-Difference Ground-Water Flow Model provides the opportunity to simulate axisymmetric flow to a well. The theory involves the conceptualization of a system of concentric shells that are capable of reproducing the large variations in gradient in the vicinity of the well by decreasing their area in the direction of the well. The computer program documented in this report serves as a preprocessor to the U.S. Geological Survey model by creating the input data file needed to implement the axisymmetric conceptualization. This report contains a listing of the computer code, a disk containing the computer code, and a description of the data input requirements to this preprocessor.

INTRODUCTION

The original U.S. Geological Survey (USGS) Modular Ground-Water Flow Model (McDonald and Harbaugh, 1988), known as MODFLOW, assumed that model nodes are in the center of rectangular cells and that transmissivity is constant within a cell. Based on these assumptions, the model calculates coefficients, called conductance, that are multiplied by head difference to determine flow between cells. A generalized finite-difference capability was developed for the USGS model (Harbaugh, 1992) to allow the model program to read conductance as input instead of calculating it internally under the assumption of block-centered finite-difference discretization.

This generalized finite-difference capability allows the conductances to be calculated externally using a node-centered radial discretization (as described in Bennett and others, 1990; and Reilly and Harbaugh, 1993) for input to the flow model. This report documents a preprocessor computer program that calculates conductances and storage capacities for a radial axisymmetric node-cen-

2

tered finite-difference grid. The calculated conductances and storage capacities can then be included as the input to MODFLOW to simulate axisymmetric flow to a well.

COMPUTER PROGRAM

The theory and logic of the computer program are described in Reilly and Harbaugh (1993), along with an example of the use of the preprocessor. A listing of the computer program used in Reilly and Harbaugh (1993) is provided in this document to make the program available and to provide a record of the original program. An MS-DOS1 compatable, 1.2 megabyte diskette that contains the computer readable text files is included with this report .

The files three files contained on the diskette are: 1. RAD_MOD.FOR, 2. RAD_MOD.TXT, and 3. NEUMAN.DAT. These computer readable files are listed in the three appendixes. Appendix 1 is a listing of the file RAD_MOD.FOR which is the Fortran source code of the preprocessor contained on the diskette. Appendix 2 is a listing of the file RAD_MOD.TXT which provides the input instructions for the input to the program as described in Reilly and Harbaugh (1993). Appendix 3 is a listing of the file NEUMAN.DAT which is the input data set used in the sample problem described in Reilly and Harbaugh (1993).

REFERENCES

Bennett, G.D., Reilly, T.E., and Hill, M.C., 1990, Technical training note in ground-water hydrology: Radial flow to a well: U.S. Geological Survey Water-Resources Investigations Report 89-4134, 83 p.

Harbaugh, A.W., 1992, A generalized finite-difference formulation for the U.S. Geological Survey modular three-dimensional finite-difference ground-water flow model: U. S. Geological Survey Open-File Report 91-494, 60 p.

McDonald, M.G., and Harbaugh, A.W., 1988, A modular three-dimensional finite-difference ground-water flow model: Techniques of Water-Resources Investigations of the U.S. Geological Survey, book 6, chap. A1, 586 p.

Reilly, T.E., and Harbaugh, A.W., 1993, Simulation of cylindrical flow to a well using the U.S.

Geological Survey modular finite-difference ground-water flow model: Ground Water, v. 31, no. 3, p.489-494.

1. The use of brand names in this report is for identification purposes only and does not constitute endorsement by the U.S. Geological Survey.

3

APPENDIXES

Appendix 1 -- Program listing

C     PREPROCESSOR TO CALCULATE CONDUCTANCES AND STORAGE CAPACITIES
C     FOR A RADIAL SIMULATION USING THE GENERAL FINITE-DIFFERENCE
C     (GFD) PACKAGE OF THE U.S.G.S. MODULAR MODEL
C
C            BY THOMAS E. REILLY AND ARLEN W. HARBAUGH
C                    VERSION 1.0  -- APRIL 1992
C
      PARAMETER (MXCOL=200,MXROW=100)
      DIMENSION R(MXCOL),AREA(MXCOL),DELZ(MXROW),KR(MXROW),KZ(MXROW),
     1        SS(MXROW),SC(MXROW,MXCOL),CR(MXROW,MXCOL),CC(MXROW,MXCOL)
      CHARACTER*80 IFNAME,OFNAM1,OFNAM2
      REAL*4 KR,KZ
      PI=3.14159
C
C  Ask for file names and open the files
      INUNIT=10
      IOUTU1=11
      IOUTU2=12
      WRITE(*,1)
1     FORMAT(5X,’ENTER INPUT FILE NAME’)
      READ(*,2) IFNAME
2     FORMAT(A)
      OPEN(INUNIT,FILE=IFNAME,STATUS=’OLD’)
      WRITE(*,3)
3     FORMAT(5X,’ENTER GFD PACKAGE OUTPUT FILE NAME’)
      READ(*,2) OFNAM1
      OPEN(IOUTU1,FILE=OFNAM1)
      WRITE(*,4)
4     FORMAT(5X,’ENTER OUTPUT FILE NAME FOR LATERAL GRID SPACING’)
      READ(*,2) OFNAM2
      OPEN(IOUTU2,FILE=OFNAM2)
C
C  Read first record, and check that arrays are dimensioned large enough
      READ(INUNIT,10) ALPHA,RW,NCOL,NROW,LOCAT
10    FORMAT(2F10.0,3I5)
      IF(NCOL.GT.MXCOL .OR. NROW.GT.MXROW) THEN
        WRITE(*,*) ‘ NCOL GREATER THAN MXCOL OR NROW GREATER THAN MXROW’
        STOP
      END IF
C
C  Calculate lateral node location
      R(1)=RW
      DO 11 I=2,NCOL
        R(I)=ALPHA*R(I-1)
11    CONTINUE
C
C  Calculate cell area
      ASQ=ALPHA*ALPHA

4

      RINTER=ALPHA*RW*SQRT(2./(ASQ+1.))
      AREA(1)=PI*(RINTER*RINTER-RW*RW)
      AREA(2)=AREA(1)*ASQ*2.
      NNC=NCOL-1
      DO 12 I=3,NNC
        AREA(I)=AREA(I-1)*ASQ
12    CONTINUE
      AREA(NCOL)=AREA(NNC)*ASQ/2.
C
C  READ VERTICAL CHARACTERISTICS
      DO 25 J=1,NROW
        READ(INUNIT,21) IR,DELZ(IR),KR(IR),KZ(IR),SS(IR)
21      FORMAT(I5,4F10.0)
25    CONTINUE
C
C  GENERATE CR, CC, AND SC ARRAYS
      DO 30 J=1,NROW
      DO 29 I=1,NCOL
        SC(J,I)=SS(J)*AREA(I)*DELZ(J)
        IF(I.EQ.NCOL) THEN
          CR(J,I)=0.0
        ELSE
          CR(J,I)=2.*PI*KR(J)*DELZ(J)/ALOG(ALPHA)
        END IF
        IF(J.EQ.NROW) THEN
          CC(J,I)=0.0
        ELSE
          A=2.*KZ(J)*AREA(I)/DELZ(J)
          IF(J.EQ.1) A=KZ(J)*AREA(I)/DELZ(J)
          B=2.*KZ(J+1)*AREA(I)/DELZ(J+1)
          IF(J+1.EQ.NROW) B=KZ(J+1)*AREA(I)/DELZ(J+1)
          C=1./A + 1./B
          CC(J,I)=1./C
        END IF
29    CONTINUE
30    CONTINUE
C
C  WRITE GFD FILE
C    ISS,IGFDCB
      WRITE(IOUTU1,41)
41    FORMAT(9X,’0’,9X,’0’,2X, ‘ISS,IGFDCB’)
C    LAYCON
      WRITE(IOUTU1,42) 0
42    FORMAT(I2)
C    DELR
      WRITE(IOUTU1,43) LOCAT,’DELR’
43    FORMAT(I10,7X,’1.0’,1x,’(5F15.0)’,20X,’0’,2X,A)
      WRITE(IOUTU1,44) (AREA(I),I=1,NCOL)
44    FORMAT(1P,5E15.5)
C    DELC
      WRITE(IOUTU1,43) LOCAT,’DELC’
      WRITE(IOUTU1,44) 1.,(0.0,I=2,NROW)
C    SC
      WRITE(IOUTU1,43) LOCAT,’SC’

5

      DO 50 J=1,NROW
      WRITE(IOUTU1,44) (SC(J,I),I=1,NCOL)
50    CONTINUE
C    CR
      WRITE(IOUTU1,43) LOCAT,’CR’
      DO 52 J=1,NROW
      WRITE(IOUTU1,44) (CR(J,I),I=1,NCOL)
52    CONTINUE
C    CC
      WRITE(IOUTU1,43) LOCAT,’CC’
      DO 53 J=1,NROW
      WRITE(IOUTU1,44) (CC(J,I),I=1,NCOL)
53    CONTINUE
C
C  Write lateral node locations
      WRITE(IOUTU2,60)
60    FORMAT(1X,’Lateral node location:’)
      WRITE(IOUTU2,44) (R(I),I=1,NCOL)
C
      STOP
      END

6

Appendix 2 --Preprocessor input instructions

INPUT TO PROGRAM:

Data set 1:

ALPHA,RW,NCOL,NROW,LOCAT

Format (2F10.0,3I5)

Data set 2 (one record for each layer in the model):

IR,DELZ(IR),KR(IR),KZ(IR),SS(IR)

Format (I5,4F10.0)

Where:

ALPHA=R(I+1)/R(I), This is the ratio of the lateral distance between neighboring nodes {usually between 1.1 and 2.0},

RW=Radius of the well

NCOL=Number of nodes laterally (variable NCOL in MODFLOW) {must be greater than 4 and is usually greater than 15}

NROW=Number of layers (variable NROW in MODFLOW)

LOCAT = Unit number for reading GFD data in MODFLOW

IR=Layer number

DELZ=Thickness of layer IR

KR=Radial hydraulic conductivity in layer IR

KZ=Vertical hydraulic conductivity in layer IR

SS=Specific storage of material in layer IR (L-1), can be set to Sy/DELZ for the top layer to represent the water table, where Sy is the specific yield (dimensionless).

Note: The output from this program is all of the required input data for the General Finite-Differ-ence Package (GFD). The value for IGFDCB (Harbaugh, 1992, p.13) is specified as zero. This can be changed, if desired, with an editor after the GFD data file has been created by the preprocessor.

7

Appendix 3 -- Preprocessor input data for sample problem

      1.50      .936   40   11   12
    1        5.     100.0      10.0 .04
    2       10.     100.0      10.0 .000005
    3       10.     100.0      10.0 .000005
    4       10.     100.0      10.0 .000005
    5       10.     100.0      10.0 .000005
    6       10.     100.0      10.0 .000005
    7       10.     100.0      10.0 .000005
    8       10.     100.0      10.0 .000005
    9       10.     100.0      10.0 .000005
   10       10.     100.0      10.0 .000005
   11        5.     100.0      10.0 .000005